
1. Przedmowa

Niniejsze opracowanie jest efektem pracy kilku osób skupionych wokół forum sp-hm.pl. 
Sterownik tutaj opisywany jest odpowiedzią na zapotrzebowanie kolegów na prosty w pełni otwarty 
projekt sterownika do nieskomplikowanych przemienników fonicznych w paśmie 2m i 70cm.

2. Informacje Licencyjne

HMGRC v.1.0
Copyright (C) 2010 Home Made Group
Niniejsze opracowanie i wszystkie jego elementy są wolnym rozwiązaniem; możesz je rozprowadzać 
dalej i/lub modyfikować na warunkach Powszechnej Licencji Publicznej GNU, wydanej przez Fundację 
Wolnego Oprogramowania - według wersji 2-giej tej Licencji lub którejś z późniejszych wersji. 
Niniejszy  opracowanie  rozpowszechniane  jest  z  nadzieją,  iż  będzie  ono  użyteczne  -  jednak  BEZ 
JAKIEJKOLWIEK  GWARANCJI,  nawet  domyślnej   gwarancji  PRZYDATNOŚCI  HANDLOWEJ  albo 
PRZYDATNOŚCI DO OKREŚLONYCH ZASTOSOWAŃ. 
W celu uzyskania bliższych informacji - Powszechna Licencja Publiczna GNU. 
Z pewnością wraz z niniejszym programem otrzymałeś też egzemplarz Powszechnej Licencji Publicznej 
GNU (GNU General Public License); jeśli nie - napisz do Free Software Foundation, Inc., 675 Mass Ave,  
Cambridge, MA 02139, USA.

3. Założenia i opis konstrukcji.

Konstrukcja ma za zadanie sterować pracą przemiennika na podstawie sygnału o zajętości kanału z 
radia odbiorczego. Ma umożliwiać włączenie opcji sygnalizacji końca nadawania korespondenta i 
identyfikacji przemiennika, a także ma mieć możliwość tłumienia sygnału audio dla radii z open-squelch.
W torze audio nie przewidziano wzmacniaczy i filtrów tak by umożliwić natywną pracę  D-Star.

Opcje konfigurowalne zworkami na porcie B
ZW1 - PB1 - włącz wyłącz opcję beep
ZW2 - PB2 - włącz wyłącz identyfikację CW po każdym TX
ZW3 - PB3 - włącz wyłącz identyfikację (włącz wyłącz automat co 5minut, jeśli wyłączone nie działa 
też po każdym TX)
ZW4 - PB4 - 1S czas podtrzymania nośnej
ZW5 - PB5 - 2S czas podtrzymania nośnej
ZW6 - PB6 - 4S czas podtrzymania nośnej
Czasy podtrzymania nośnej się sumują więc można ustawić czas od 1S do 7S ze skokiem co 1S. 
Dodatkowo zdjęcie wszystkich zworek od PB4 - PB6 ustawia czas podtrzymania nośnej na 10S
ZW7 – PB7 – Konfiguracja poziomu SQL 0V/5V

Piny:
PD5 - Sygnalizacja TX
PD4 - Sygnalizacja RX

http://sp-hm.pl/


4. Schemat elektryczny

5. Lista części.

| C1         100uF       
| C2         100nF       
| C3         33          
| C4         33          
| C5         1nF         
| C6         470nF       
| C7         47nF        
| C8         470nF       
| C9         100nF       
| C10        100uF       
| D1         LED RX      
| D2         LED TX      
| PR1        22k         
| PR2        22k         
| R1         2k2         
| R2         10k         
| R3         22K         
| R4         2k2         
| R5         2k2         
| R6         10k         
| R7         2k2         
| R8         4k7         
| R9         2k2         
| R10        470         
| R11        470         
| R12        4k7         
| R13        10k         
| T1         BC547       
| T2         BC547       
| T3         BC547       
| T4         BC547       
| U1         ATTINY2313  
| U2         7805        
| X1         4MHz        
   



6. Rysunki płytek

7. Kod źródłowy



$regfile = Attiny2313.dat
$crystal = 4000000
$baud = 9600
$hwstack = 32
$swstack = 0
$framesize = 12

'wejscia konfiguracyjne port b portb.0 jako wyjscie PTT
'wejsca sygnalowe i wyjscia sygnalizacyjne
Config Portb = &B10000001
Config Portd = &B1111101

'ustawiamy wysokie stany na wejsciach i niskie na wyjściach
Portb = &B11111110
Portd = &B0000000

'aliasy na porcie D
Sql Alias Pind.1
Morsepin Alias Portd.6
Tx_led Alias Portd.5
Rx_led Alias Portd.4
Mute Alias Portd.3
Wentylator Alias Portd.2

'aliasy na porcie B
Ptt Alias Portb.0
Beep_option_jumper Alias Pinb.1
Ident_after_jumper Alias Pinb.2
Ident_option_jumper Alias Pinb.3
Podtrzymaniea Alias Pinb.4
Podtrzymanieb Alias Pinb.5
Podtrzymaniec Alias Pinb.6
Sql_lvl Alias Pinb.7

'zegar on
Enable Interrupts
Config Timer1 = Timer , Prescale = 64
Enable Timer1

'definicje stałych
Const Speed = 12
Const Czestotliwosc = 450
Const Kropka = Czestotliwosc / Speed
Const Kreska = 3 * Kropka

'definicje zmiennych
Dim Sendchar As String * 1
Dim Charcode As Byte
Dim Dataindex As Byte
Dim Charindex As Byte
Dim Charstring As String * 12
Dim Stringlength As Byte
Dim Codelength As Byte
Dim Codeindex As Byte
Dim Codeelement As String * 1
Dim Sendcw As Bit
Dim Rx_flag As Bit
Dim Podtrzymanie_flag As Bit
Dim Rx As Bit
Dim Rxa As Bit
Dim Sekunda As Byte
Dim Minuta As Byte
Dim Countdown As Byte
Dim Inputtekst As String * 6
Dim Sendcwinterwal As Byte
Dim Czaspodtrzymania As Byte
Dim Odtworzylem_beep As Bit

'konfiguracja zmiennych sterownika
Inputtekst = "SR5WM"     'znamiennik max 6 znaków jesli chcesz wiekszy zmien rozmiar zmiennej Inputtekst
Sendcwinterwal = 10      'czas automatycznego wysyłania znamiennika w minutach

'-------------------------------------------------------------------------------
'tutaj uruchamiam timer do odliczania wyłączania flag
On Timer1 Liczmy
Wait 3

'Uruchomienie Przemiennika Wysyłamy znamiennik
Sendcw = 1

'główna pętla programu
'-------------------------------------------------------------------------------
Do
'ustawiam czas podtrzymania ze zworek
If Podtrzymaniea = 0 And Podtrzymanieb = 0 And Podtrzymaniec = 0 Then
   Czaspodtrzymania = 7  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 1 And Podtrzymanieb = 0 And Podtrzymaniec = 0 Then
   Czaspodtrzymania = 6  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 0 And Podtrzymanieb = 1 And Podtrzymaniec = 0 Then
   Czaspodtrzymania = 5  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 1 And Podtrzymanieb = 1 And Podtrzymaniec = 0 Then
   Czaspodtrzymania = 4  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 0 And Podtrzymanieb = 0 And Podtrzymaniec = 1 Then
   Czaspodtrzymania = 3  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 1 And Podtrzymanieb = 0 And Podtrzymaniec = 1 Then
   Czaspodtrzymania = 2  'podstawowy czas podtrzymania nosnej w sekundach
Elseif Podtrzymaniea = 0 And Podtrzymanieb = 1 And Podtrzymaniec = 1 Then
   Czaspodtrzymania = 1  'podstawowy czas podtrzymania nosnej w sekundach
Else
   Czaspodtrzymania = 10



End If

'obsługa sygnału SQL zapalanie flagi RX
'-------------------------------------------------------------------------------
'jesli pin SQL ma stan inny niż zworka ZW7
   If Sql <> Sql_lvl Then
'ustawiam bit pomocniczy
         Rx = 1
'jeśli po 20ms nadal wciśnięty uruchamiam zatrzask Rxa
      If Rx = 1 And Rxa = 0 Then
         Rxa = 1
'ustawiamy flage jest RX
         Rx_flag = 1
      End If
'jeśli sygnał zajętości znika to
   Elseif Rxa = 1 Then
         Rx_flag = 0
         Rx = 0 : Rxa = 0
      End If

'ustawianie flagi TX jesli ustawione flagi RX lub podtrzymanie
'-------------------------------------------------------------------------------

'ustawianie flagi podtrzymanie jesli zapalony RX
If Rx_flag = 1 Then
         Podtrzymanie_flag = 1
         Odtworzylem_beep = 0
         Countdown = 0
         Rx_led = 1
         Mute = 0
Else

         Rx_led = 0
         Mute = 1
End If

'ustawiamy port PTT i inne jesli mamy flage podtrzymanie
 If Podtrzymanie_flag = 1 Then
 'wysterowanie nadajnika
   Ptt = 1
   Tx_led = 1
   Wentylator = 1
 'tutaj beep jesli koniec nadawania i beep_option_jumper enabled  (0 to stan niski zworka założona)
 If Rx_flag = 0 And Beep_option_jumper = 0 And Odtworzylem_beep = 0 Then
   Waitms 300
   'sprawdzic czy stoi nośna jeśli nie to beep
    If Sql = Sql_lvl Then
      Sound Morsepin , Kropka , Czestotliwosc
      Odtworzylem_beep = 1
    End If
   End If
 Else
 'wysterowanie nadajnika
    Ptt = 0
    Tx_led = 0
    Wentylator = 0
 End If

'procedura wysylania znamiennika
'-------------------------------------------------------------------------------
'ident option jesli zwarty do masy to będzie nadawany znamiennik
If Sendcw = 1 And Ident_option_jumper = 0 Then
'uruchamiam sobie TX
Ptt = 1
Tx_led = 1
'blokujemy timer na czas wysylania znamiennikabo chrypi
 Stop Timer1
 Waitms 300
   Stringlength = Len(inputtekst)
'tutaj przesuwamy się po ciągu znaków i będziemy odtwarzać każdą literę z osobna
   For Charindex = 1 To Stringlength
   Sendchar = Mid(inputtekst , Charindex , 1)
   Charcode = Asc(sendchar)

    Dataindex = Charcode - 48
    Charstring = Lookupstr(dataindex , Alfabet)
    Codelength = Len(charstring)

'ta pętla generuje kazda litere znaku
         For Codeindex = 1 To Codelength
               Codeelement = Mid(charstring , Codeindex , 1)
            If Codeelement = "." Then
'Zagraj kropke
               Sound Morsepin , Kropka , Czestotliwosc
            Else
'Zagraj kreske
               Sound Morsepin , Kreska , Czestotliwosc
            End If
            If Codeindex < Codelength Then
               Waitms Kropka
            End If
         Next Codeelement
'odstęp pomiedzy znakami 3 x kropka
      Waitms Kreska
'i wracamy do początku pętli po następną literę
   Next Charindex
'skonczylismy gasimy flaga nadawania znamiennika
'resetujemy lflagi
   Sendcw = 0
   Ptt = 0



   Sekunda = 0
'skonczylismy nadawac znamiennik odpalamy timer 1
   Start Timer1
   End If

'koniec pętli
Loop
'koniec programu
End

'obsługa
'-------------------------------------------------------------------------------

Liczmy:
Timer1 = 0
'tutaj proste naliczanie w sekundzie i cos (dla uproszczenia i skrocenia kodu uzyłem timera tc1/64 to daje 1s = 1,05S)
   Sekunda = Sekunda + 1
   If Sekunda = 60 Then
      Minuta = Minuta + 1
      If Minuta = 60 Then
         Minuta = 0
      End If
      Sekunda = 0
   End If

'jesli podtrzymanie na 1 a rx spadl zczynamy odliczanie do wylaczenia
   If Podtrzymanie_flag = 1 And Rx_flag = 0 Then
     Countdown = Countdown + 1
'jesli doszlismy do konca gas flagi
      If Countdown >= Czaspodtrzymania Then
         Podtrzymanie_flag = 0
         Ptt = 0
         Tx_led = 0
'jesli zaznaczona flaga identylfikacji po odpuszczeniu tx to zagraj znamiennik
         If Ident_after_jumper = 0 Then
            Minuta = Sendcwinterwal
         End If
      End If
'reset licznika jesli pojawi sie flaga rx w trakcie podtrzymania
   Elseif Podtrzymanie_flag = 1 And Rx_flag = 1 Then
      Countdown = 0
   End If

'okresowe wysyłanie znamiennika jesli przekroczono czas i nikt nie nadaje
   If Minuta >= Sendcwinterwal And Rx_flag = 0 And Ident_option_jumper = 0 Then
        Sendcw = 1
        Minuta = 0
   End If

Return
'-------------------------------------------------------------------------------

Alfabet:
Data "-----"             '0  Znaki pokolei tak jak są ustawione w tablicy ASCII
Data ".----"             '1  przy wywoływaniu musimy tylko przestawić indeks o 48
Data "..---"             '2  ponieważ cyfra 0 ma kod ASCII 48
Data "...--"             '3
Data "....-"             '4
Data "....."             '5
Data "-...."             '6
Data "--..."             '7
Data "---.."             '8
Data "----."             '9
Data "."                 ': do generowania znamienników znaki specjalne nie są potrzebne
Data "."                 ';
Data "."                 '<
Data "."                 '=
Data "."                 '>
Data "."                 '?
Data "."                 '@
Data ".-"                'A  Literki
Data "-..."              'B
Data "-.-."              'C
Data "-.."               'D
Data "."                 'E
Data "..-."              'F
Data "--."               'G
Data "...."              'H
Data ".."                'I
Data ".---"              'J
Data "-.-"               'K
Data ".-.."              'L
Data "--"                'M
Data "-."                'N
Data "---"               'O
Data ".--."              'P
Data "--.-"              'Q
Data ".-."               'R
Data "..."               'S
Data "-"                 'T
Data "..-"               'U
Data "...-"              'V
Data ".--"               'W
Data "-..-"              'X
Data "-.--"              'Y
Data "--.."              'Z

7. Twórcy



SQ9MDD Ryszard Koncepcja, oprogramowanie, rysunki płytek.
SQ5MGE Sebastian Opracowanie elektroniczne
SQ5WPJ Paweł Konsultacje merytoryczne
SP5FCS Adam Rysunki płytek
3Z6AEF Waldek Rysunki płytek


